Maple and the Putnam Competition

Philippe Dumas and Bruno Salvy *

Introduction

The Putnam Competition is a yearly mathematical contest at the advanced undergraduate level. The problems can generally be solved in an astute manner which does not require heavy computation or advanced mathematical knowledge. Our approach here is completely different. We show that even without subtlety, brute-force attacks can solve many of these problems, assisted by the computing power provided by modern computer algebra systems. Both approaches are complementary, since in real life computer algebra is often used in the experimentation stage before reaching general mathematical results.

We have selected the 1993 Competition as an example. We do not have any reason to think that this was a particularly simple year. We now proceed to treat each of these problems in turn. For some of them (A-3, A-4, B-2, B-4, B-6), we did not find any way in which computer algebra could help find the solution. In some of these cases, we shall explain this failure. Among the twelve problems of this contest, Maple can be used to solve six and give a strong indication for a seventh one.

Problem A-1



The horizontal line $y = c$ intersects the curve $y = 2x - 3x^3$ in the first quadrant as in the figure. Find c so that the areas of the two shaded regions are equal.

Denoting by a and b the abscissæ of the intersections of $y = c$ with the curve $y = 2x - 3x^3$, this problem is easily translated into a system of equations:

*Algorithms Project, INRIA, 78153 Le Chesnay Cedex, France. `Philippe.Dumas@inria.fr` and `Bruno.Salvy@inria.fr`

```
> {a*c-int(2*x-3*x^3,x=0..a)=
>   int(2*x-3*x^3,x=a..b)-(b-a)*c,
> 2*a-3*a^3=c,2*b-3*b^3=c};
```

$$\Big\{2\,a - 3\,a^3 = c, 2\,b - 3\,b^3 = c,\\ a\,c - a^2 + \frac{3}{4}\,a^4 =\\ b^2 - \frac{3}{4}\,b^4 - a^2 + \frac{3}{4}\,a^4 - (b-a)\,c\Big\}$$

Now, we just solve this system:

```
> solve(",{a,b,c});
```

$$\{a = 0, b = 0, c = 0\}, \{\\ a = \mathrm{RootOf}(3\,\_Z^2 - 2), b = 0,\\ c = 0\}, \left\{b = \frac{2}{3}, a = \frac{2}{3}, c = \frac{4}{9}\right\}, \Big\{\\ b = \frac{2}{3},\\ a = \mathrm{RootOf}(9\,\_Z^2 + 6\,\_Z - 2),\\ c = \frac{4}{9}\Big\}, \left\{a = \frac{-2}{3}, c = \frac{-4}{9}, b = \frac{-2}{3}\right\}, \Big\{\\ a = \mathrm{RootOf}(9\,\_Z^2 - 6\,\_Z - 2),\\ c = \frac{-4}{9}, b = \frac{-2}{3}\Big\}$$

Only one of these solutions satisfies $0 < a < b$. This gives the result $c = 4/9$.

Problem A-2

Let $(x_n)_{n\geq 0}$ be a sequence of nonzero real numbers such that

$$x_n^2 - x_{n-1}x_{n+1} = 1, \qquad \textit{for } n = 1, 2, 3, \ldots$$

Prove there exists a real number a such that $x_{n+1} = ax_n - x_{n-1}$ for all $n \geq 1$.

This is reminiscent of an identity of Cassini:

$$F_n^2 - F_{n-1}F_{n+1} = (-1)^n,$$

where F_n denotes the nth Fibonacci number, of which the problem is actually a simple generalization.

The sequence (x_n) is completely determined by its two first values x_0 and x_1. Defining the sequence (y_n) by

$$y_{n+1} = ay_n - y_{n-1}, \quad y_0 = x_0, \quad y_1 = x_1,$$

we shall prove that $y_n = x_n$ for all n by showing that (y_n) satisfies the same recurrence as (x_n) for some particular value of a.

We use the gfun package from the share library[1] [1]. This package provides procedures to compute linear recurrences satisfied by sums or products of sequences themselves defined by linear recurrences.

We first define the sequence (y_n):

```
> r:=y(n+1)=a*y(n)-y(n-1):
> ini:=y(0)=x[0],y(1)=x[1]:
```

This recurrence is also satisfied by (y_{n-1}) and (y_{n+1}), with different initial conditions. A recurrence satisfied by (y_n^2) is given by

```
> `rec*rec`({r,ini},{r,ini},y(n));
```

$$\{\mathrm{y}(3) = (-c_1 - a\,c_0 + a^2\,c_1)^2, \mathrm{y}(2) = (-c_0 + a\,c_1)^2, \mathrm{y}(0) = {c_0}^2, \mathrm{y}(1) = {c_1}^2, \mathrm{y}(n) + (-2 + 3\,a^2 - a^4)\,\mathrm{y}(n+2) + (-2\,a^2 + a^4)\,\mathrm{y}(n+3) + (1 - a^2)\,\mathrm{y}(n+4)\}$$

While gfun insists on using the same name $y(n)$ in this recurrence as in the original one, there should be no confusion that the above recurrence is actually satisfied by (y_n^2). Similarly, we obtain a recurrence satisfied by $(y_{n+1}y_{n-1})$ and a recurrence satisfied by the difference of these sequences.

```
> `rec*rec`({r,y(0)=x[1],y(1)=a*x[1]-x[0]},
>            {r,y(1)=x[0],y(2)=x[1]},y(n)):
> `rec+rec`(`rec*rec`(y(n)=-1,",y(n)),"",y(n));
```

$$\{\mathrm{y}(7) = \%1, \mathrm{y}(5) = \%1, \mathrm{y}(6) = \%1, \mathrm{y}(4) = \%1, \mathrm{y}(2) = \%1, \mathrm{y}(0) = \%1, \mathrm{y}(1) = \%1, \mathrm{y}(3) = \%1, (-1 + a^2)\,\mathrm{y}(n) + (-a^{10} + 4\,a^8 - 4\,a^6)\,\mathrm{y}(n+7) + (a^8 - 2\,a^6 - a^4 + 3\,a^2 - 1)\,\mathrm{y}(n+8) + (-a^{10} + 7\,a^8 - 20\,a^6 + 25\,a^4 - 12\,a^2 + 2)\,\mathrm{y}(n+4) + (2\,a^{10} - 12\,a^8 + 26\,a^6 - 24\,a^4 + 8\,a^2)\,\mathrm{y}(n+5)\}$$
$$\%1 := -c_0\,a\,c_1 + {c_1}^2 + {c_0}^2$$

[1] An updated version of the gfun package that was used in this paper is available in the electronic updates to the share library, which are accessible via anonymous ftp. The latest version can be obtained by anonymous ftp from `ftp.inria.fr:INRIA/Projects/algo/programs/gfun`.

The conclusion is obtained by checking that the sequence $y_n = 1$ is a solution to this recurrence for some value of a:

```
> eval(subs(y=1,"));
```

$$\{0, 1 = -x_0 a x_1 + {x_1}^2 + {x_0}^2\}$$

Another approach to this problem consists in using the theory of rational sequences. In this context, the result is obtained via properties of Hankel determinants (see [2, pp. 96–100]).

Problem A-3

Let $\mathcal{P}_n$ be the set of subsets of $\{1, 2, \ldots, n\}$. Let $c(n, m)$ be the number of functions $f : \mathcal{P}_n \to \{1, 2, \ldots, m\}$ such that $f(A \cap B) = \min\{f(A), f(B)\}$. Prove that

$$c(n, m) = \sum_{j=1}^{m} j^n.$$

This problem is not very difficult. These functions are completely determined by their values on $\{1, 2, \ldots, n\} \setminus \{i\}$, for $i = 1, 2, \ldots, n$.

Problem A-4

Let $x_1, x_2, \ldots, x_{19}$ be positive integers each of which is less than or equal to 93. Let $y_1, y_2, \ldots, y_{93}$ be positive integers each of which is less than or equal to 19. Prove that there exists a (nonempty) sum of some x_i's equal to a sum of some y_j's.

The question does not actually depend on the particular numbers 19 and 93. We can use the following procedure to check the property for particular values. It takes as input two positive integers m and n. For each n-tuple of integers in $\{1, \ldots, m\}$ and for each m-tuple of integers in $\{1, \ldots, n\}$ it checks whether the sets of partial sums of both have a nonempty intersection. This is done using the iterator cartprod from the combinat package, which avoids generating the whole sets of possibilities and thus makes the program run with very small memory requirements.

```
> check:=proc(m,n)
> local X,x,Y,y,i,s;
>   X:=cartprod([seq([$1..m],i=1..n)]):
>   while not X[finished] do
>     x:=X[nextvalue]();
>       s:={op(map(convert,
>           powerset(x),`+`))};
>       Y:=cartprod([seq([$1..n],i=1..m)]);
>       while not Y[finished] do
>         y:=Y[nextvalue]();
>         if s intersect
>          {op(map(convert,powerset(y),`+`))}
>          minus {0}={} then
```

```
>             print(x,y);
>             RETURN(false)
>         fi;
>       od;
>   od;
>   true
> end:
```

For instance, with 3 and 4 we get:

```
> check(3,4);
```

$$true$$

Unfortunately, the combinatorial explosion of the number of cases to consider makes it impossible to apply this procedure to 19 and 93. A clever pigeon-hole argument is given in [3].

Problem A-5

Show that

$$\int_{-100}^{-10}\left(\frac{x^2-x}{x^3-3x+1}\right)^2 dx + \int_{\frac{1}{101}}^{\frac{1}{11}}\left(\frac{x^2-x}{x^3-3x+1}\right)^2 dx + \int_{\frac{101}{100}}^{\frac{11}{10}}\left(\frac{x^2-x}{x^3-3x+1}\right)^2 dx$$

is a rational number.

Definite integrals of rational functions are very well understood and it comes as a surprise that Maple is not able to compute the result without help. We therefore start from the indefinite integral.

```
> f:=((x^2-x)/(x^3-3*x+1))^2:
> g:=int(f,x);
```

$$g := \frac{-\frac{2}{9}+\frac{8}{9}x-\frac{5}{9}x^2}{x^3-3\,x+1}+\left(\sum_{\_R=\%1}\_R\,\ln\left(x-\frac{27}{2}\_R\right)\right)$$

$$\%1 := \mathrm{RootOf}(\,19683\_Z^3-324\_Z+8\,)$$

As follows from the general theory of integration of rational functions, this is the sum of a rational function and a logarithmic part. The sum of logarithms is indexed by algebraic quantities that are the residues of the rational function. The question then amounts to showing that this sum vanishes for the special values of the endpoints of the integrals. We start by isolating the logarithms associated with each residue and evaluating them at the endpoints.

```
> lnpart:=op(2,g):
> L:=op(1,lnpart):
> h:=subs(x=-10,L)-subs(x=-100,L)
>   +subs(x=1/11,L)-subs(x=1/101,L)
>   +subs(x=11/10,L)-subs(x=101/100,L);
```

$$h := \_R\ln\left(-10-\frac{27}{2}\_R\right)-\_R\ln\left(-100-\frac{27}{2}\_R\right)+\_R\ln\left(\frac{1}{11}-\frac{27}{2}\_R\right)-\_R\ln\left(\frac{1}{101}-\frac{27}{2}\_R\right)+\_R\ln\left(\frac{11}{10}-\frac{27}{2}\_R\right)-\_R\ln\left(\frac{101}{100}-\frac{27}{2}\_R\right)$$

```
> h:=combine(factor(h),ln);
```

$$h := \_R\ln\left(\left(-10-\frac{27}{2}\_R\right)\left(\frac{1}{11}-\frac{27}{2}\_R\right)\left(\frac{11}{10}-\frac{27}{2}\_R\right)\Big/\left(\left(-100-\frac{27}{2}\_R\right)\left(\frac{1}{101}-\frac{27}{2}\_R\right)\left(\frac{101}{100}-\frac{27}{2}\_R\right)\right)\right)$$

The result follows from noticing that this logarithm is actually independent of the particular residue and the fact that the sum of residues is 0. We first replace _R by its value inside the logarithm and normalize using `evala`.

```
> evala(subs(op(2,lnpart),op(2,h)));
```

$$\ln\left(\frac{326230}{3665563}\right)$$

We then sum over all the residues.

```
> sum(_R*",op(2,lnpart));
```

$$0$$

Note that it is not easy to generate such an example of a rational sum of three integrals of a rational function with an irreducible denominator of degree 3. As F. Morain pointed out to us, the reason why this example works is that the splitting field of $f(x)=x^3-3x+1$ is a cyclic cubic field. Thus, there exists an homography σ (in this example $t\mapsto 1-1/t$) which represents a generator of the Galois group of the polynomial over $\mathbf{Q}$. As a result, for any root α of $f(x)$, the product

$$(t-\alpha)(\sigma(t)-\alpha)(\sigma^2(t)-\alpha)$$

factors as $f(t)g(\alpha)/d(t)$, with $d\in\mathbf{Q}[t]$ and $g\in\mathbf{Q}[\alpha]$. This is why the logarithm above is independent of the particular residue. Ultimately, this property is equivalent to the discriminant of the cubic polynomial $f(x)$ being a perfect square (here 81) [4, p. 330].

Problem A-6

The infinite sequence of 2's and 3's

$$2,3,3,2,3,3,3,2,3,3,3,2,3,3,2,3,3,\\ 3,2,3,3,3,2,3,3,3,2,3,3,2,3,3,3,2,\dots$$

has the property that, if one forms a second sequence that records the number of 3's between successive 2's, the result is identical to the given sequence. Show that there exists a real number r such that, for any n, the nth term of the sequence is 2 if and only if $n = 1 + \lfloor rm \rfloor$ for some nonnegative integer m. (Note: $\lfloor x \rfloor$ denotes the largest integer less than or equal to x.)

When read backwards, the property defining the sequence gives a powerful way of generating the sequence:

```
> l[0]:=[2]:
> for i to 10 do
>   l[i]:=subs([2=(2,3,3),3=(2,3,3,3)],
>         l[i-1])
> od:
```

```
> l[3];
```

$$[2,3,3,2,3,3,3,2,3,3,3,2,3,3,2,\\ 3,3,3,2,3,3,3,2,3,3,3,2,3,\\ 3,2,3,3,3,2,3,3,3,2,3,3,3]$$

If a real number r satisfying the conditions of the question exists, then the proportion of 2's in the sequence must tend to $1/r$. We shall use this remark to find a good candidate for r.

```
> select(type,l[10],identical(2)):
> r:=nops(l[10])/nops(");
```

$$r := \frac{413403}{110771}$$

The partial fraction expansion of this number turns out to be very simple:

```
> convert(r,confrac);
```

$$[3,1,2,1,2,1,2,1,2,1,2,1,2,1,2,1,2]$$

The periodicity indicates that the number is likely to be the root of a quadratic polynomial. This polynomial in turn can be guessed by Maple's function `minpoly`.

```
> readlib(lattice): minpoly(r,2);
```

$$1 - 4\_X + \_X^2$$

```
> solve(",_X);
```

$$2+\sqrt{3}, \quad 2-\sqrt{3}$$



Figure 1: *The line $y = (\sqrt{3}-1)x$ encodes the sequence 2,3,3,2,3,...*

We now check that the number r satisfies the desired property up the the 1000th term of the sequence.

```
> r:=2+sqrt(3):
> n:=0:
> for m from 0 while n<1000 do
>     s:=1+trunc(r*m);
>     for n from n+1 to s-1 do
>       l2[n]:=3
>     od;
>     l2[s]:=2;
>   od:
> L:=l[6]:
> evalb([seq(L[i],i=1..1000)]
>       =[seq(l2[i],i=1..1000)]);
```

true

Once r has been guessed (as above), to complete the proof requires computing inequalities to prove an induction about the location of the indices of the 2's (see [3]).

This problem is related to morphisms in the theory of formal languages and admits a geometrical interpretation. Letting $\{r\}$ denote the fractional part of r, the sequence $2,3,3,2,\ 3,\cdots$ can be obtained by drawing in the plane the line $y=\{r\}x$ and writing a letter 3 at position n if the line crosses a horizontal line $y=k$ ($k \in N$) between $x=n$ and $x=n+1$ and a letter 2 otherwise (see Fig. 1 and [5, 6]). We thank J. Berstel for clarifying this connection for us.

Problem B-1

Find the smallest positive integer n such that for every integer m, with $0 < m < 1993$, there exists an integer k for which

$$\frac{m}{1993} < \frac{k}{n} < \frac{m+1}{1994}.$$

Here pure brute force is sufficient:

```
> for n do
>   for m to 1992 do
>     if irem((m+1)*n,1994,'k')=0
>       then k:=k-1 fi;
>     if m*n>=1993*k then break fi
>   od;
>   if m=1993 then print(n); break fi
>   od;
```

$$3987$$

The result turns out to be $1993+1994$ which could have been obtained directly by considerations similar to those used in the study of Farey series (see [7]).

Problem B-2

Consider the following game played with a deck of $2n$ cards numbered from 1 to $2n$. The deck is randomly shuffled and n cards are dealt to each of two players, A and B. Beginning with A, the players take turns discarding one of their remaining cards and announcing its number. The game ends as soon as the sum of the numbers on the discarded cards is divisible by $2n+1$. The last person to discard wins the game. Assuming optimal strategy by both A and B, what is the probability that A wins?

It requires some thought to realize that B can play in such a way as to win every game. Computer algebra does not help.

Problem B-3

Two real numbers x and y are chosen at random in the interval $(0,1)$ with respect to the uniform distribution. What is the probability that the closest integer to x/y is even? Express the answer in the form $r+s\pi$, where r and s are rational numbers.

The question is easily seen to be equivalent to the computation of the area of the shaded region in the following figure.



This is readily translated into

```
> res:=int(y/2,y=0..1)+
> sum(int(2*x/(4*p-1)-2*x/(4*p+1),x=0..1),
>   p=1..infinity);
```

$$res := \frac{5}{4} - \frac{1}{4}\Psi\left(\frac{3}{4}\right) + \frac{1}{4}\Psi\left(\frac{1}{4}\right)$$

and the final result is obtained by the simplification

```
> combine(res,Psi);
```

$$\frac{5}{4} - \frac{\pi}{4}$$

Problem B-4

The function $K(x,y)$ is positive and continuous for $0 \le x \le 1$, $0 \le y \le 1$, and the functions $f(x)$ and $g(x)$ are positive and continuous for $0 \le x \le 1$. Suppose that for all x, $0 \le x \le 1$,

$$\int_0^1 f(y)K(x,y)\,dy = g(x)$$

and

$$\int_0^1 g(y)K(x,y)\,dy = f(x).$$

Show that $f(x) = g(x)$ for $0 \le x \le 1$.

We could not find any solution to this problem using computer algebra.

Problem B-5

Show there do not exist four points in the Euclidean plane such that the pairwise distances between the points are all odd integers.

The distances between $n+2$ points in a Euclidean space of dimension n are not completely independent. In particular the following determinant — the Cayley-Menger determinant — must be zero [9]:

$$\begin{vmatrix} 0 & 1 & 1 & \cdots & 1 \\ 1 & 0 & d_{1,2}^2 & \cdots & d_{1,n+2}^2 \\ 1 & d_{2,1}^2 & 0 & \ddots & \vdots \\ \vdots & \vdots & \ddots & \ddots & d_{n+1,n+2}^2 \\ 1 & d_{n+2,1}^2 & \cdots & d_{n+2,n+1}^2 & 0 \end{vmatrix}.$$

Here $d_{i,j}$ denotes the Euclidean distance between points number i and j. To solve the problem, we first compute this determinant for $n=2$.

```
> nb:=4:
> CM:=array(1..nb+1,1..nb+1,
>     sparse,symmetric):
> for j from 2 to nb+1 do CM[1,j]:=1 od:
> for i from 2 to nb+1 do
>   for j from i+1 to nb+1 do
>       CM[i,j]:=d[i-1,j-1]^2
>   od od:
> print(CM);
```

$$\begin{bmatrix} 0,1,1,1,1 \\ 1,0,{d_{1,2}}^2,{d_{1,3}}^2,{d_{1,4}}^2 \\ 1,{d_{1,2}}^2,0,{d_{2,3}}^2,{d_{2,4}}^2 \\ 1,{d_{1,3}}^2,{d_{2,3}}^2,0,{d_{3,4}}^2 \\ 1,{d_{1,4}}^2,{d_{2,4}}^2,{d_{3,4}}^2,0 \end{bmatrix}$$

```
> DCM:=linalg[det](CM):
```

We then make the substitution $d_{i,j} = 2k_{i,j} + 1$ and reduce modulo 8:

```
> DCMk:=subs({seq(seq(
>    d[i,j]=2*k[i,j]+1,j=i+1..nb),
>    i=1..nb)},DCM):
> Expand(DCMk) mod 8;
```

$$4$$

This result shows that the determinant cannot be 0 when the distances are odd integers, which proves the desired result.

Problem B-6

Let S be a set of three, not necessarily distinct, positive integers. Show that one can transform S into a set containing 0 by a finite number of applications of the following rule: Select two of the three integers, say x and y, where $x \leq y$, and replace them with $2x$ and $y - x$.

Here again, we were unable to come up with a solution using Maple.

Conclusion

These examples are typical of the everyday use of Maple. For some problems, computer algebra is useless; for other problems, it provides a complete solution in one line; for most problems however, interaction with the system is necessary to develop intuition and to arrive at a more or less complete solution.

Acknowledgment

This work was supported in part by the Esprit III Basic Research Action of the EEC under contract ALCOM II (#7141).

References

[1] B. Salvy and P. Zimmermann: Gfun: a Maple package for the manipulation of generating and holonomic functions in one variable. *ACM Trans. Math. Soft. 20*, 2, pp. 163–177, (1994).

[2] G. Pólya and G. Szegö: *Problems and Theorems in Analysis II*, 4th ed. Springer-Verlag, (1976).

[3] L. F. Klosinski, G. L. Alexanderson and L. C. Larson: The fifty-fourth William Lowell Putnam mathematical competition. *Amer. Math. Monthly 101*, 8, pp. 725–734, (1994).

[4] H. Cohen: *A course in computational algebraic number theory*, vol. 138 of *Graduate Texts in Mathematics*. Springer-Verlag, (1993).

[5] J. Berstel and P. Séébold: Morphismes de Sturm. *Bull. Belgian Math. Soc. 1*, 2, pp. 175–189, (1994).

[6] T. C. Brown: Descriptions of the characteristic sequence of an irrational. *Canadian Math. Bull. 36*, 1, pp. 15–21, (1993).

[7] G. H. Hardy and E. M. Wright: *An Introduction to the Theory of Numbers*, fifth ed. Oxford University Press, (1979).

[8] E. T. Whittaker and G. N. Watson: *A Course of Modern Analysis*, fourth ed. Cambridge University Press, 1927. (Reprinted 1973).

[9] L. M. Blumenthal: *Theory and applications of distance geometry*, 2nd ed. Chelsea Pub. Co., (1970).